

TLA⁺ Proofs

Denis Cousineau¹, Damien Doligez^{1,2}, Leslie Lamport³, Stephan Merz⁴,
Daniel Ricketts⁵, and Hernán Vanzetto^{1,4}

¹ INRIA - Université Paris Sud, Orsay, France. **

² INRIA, Paris, France

³ Microsoft Research, Mountain View, CA, U.S.A.

⁴ INRIA & LORIA, Nancy, France

⁵ Department of Computer Science, University of California, San Diego, U.S.A.

Abstract. TLA⁺ is a specification language based on standard set theory and temporal logic that has constructs for hierarchical proofs. We describe how to write TLA⁺ proofs and check them with TLAPS, the TLA⁺ Proof System. We use the simple Peterson mutual exclusion algorithm as an example to describe the features of TLAPS and show how it and the Toolbox (an IDE for TLA⁺) help users to manage large, complex proofs.

1 Introduction

TLA⁺ [13] is a specification language originally designed for specifying concurrent and distributed systems and their properties. Specifications and properties are written as formulas of TLA, a linear-time temporal logic. TLA⁺ is based on TLA and Zermelo-Fraenkel set theory with the axiom of choice; it also adds a module system for structuring specifications. More recently, constructs for writing proofs have been added to TLA⁺; these are derived from a hierarchical presentation of natural-deduction proofs proposed for writing rigorous hand proofs [12].

In this paper, we present the main ideas that guided the design of the proof language and our experience with using the TLA⁺ tools for verifying safety properties of TLA⁺ specifications. The TLA⁺ Toolbox is an integrated development environment (IDE) based on Eclipse for writing TLA⁺ specifications and running the TLA⁺ tools on them, including the TLC model checker and TLAPS, the TLA⁺ proof system [5]. In particular, it provides commands to hide and unhide parts of a proof, allowing a user to focus on a given proof step and its context. It is also useful to be able to run the model checker on the same formulas that one reasons about.

The TLA⁺ proof language and TLAPS have been designed to be independent of any particular theorem prover. All interaction takes place at the level of TLA⁺, letting the user focus on the specification of the algorithm being developed. We do not expect users to have precise knowledge of the inner workings of the

** This work was partially funded by INRIA-Microsoft Research Joint Centre, France.

backend provers that TLAPS uses, although with experience users learn about the strengths and weaknesses of the different provers.

TLAPS has a *Proof Manager* (PM) that transforms a proof into individual proof obligations that it sends to backend provers. Currently, the main backend provers are Isabelle/TLA⁺, a faithful axiomatization of TLA⁺ as an object logic in Isabelle [23], Zenon [4], a tableau prover for classical first-order logic with equality, and SMT solvers (using either a generic SMT-LIB2 [3] backend or optimized translations to the native input formats of Yices [9] and Z3 [8]). Isabelle serves as the most trusted backend prover, and when possible, we expect backend provers to produce a detailed proof that is checked by Isabelle. This is currently implemented for the Zenon backend, which can export its proofs as Isar scripts that Isabelle can certify.

We explain how to write and check TLA⁺ proofs, using a tiny well-known example: a proof that Peterson’s algorithm [20] implements mutual exclusion. We start by writing the algorithm in PlusCal [14], an algorithm language that is based on the expression language of TLA⁺. The PlusCal code is translated to a TLA⁺ specification, which is what we reason about. Section 3 introduces the salient features of the proof language and of TLAPS with the proof of mutual exclusion. Liveness of Peterson’s algorithm (processes eventually enter their critical section) can also be asserted and proved with TLA⁺. However, liveness reasoning makes full use of temporal logic, and TLAPS cannot yet check temporal logic proofs. We therefore discuss only mutual exclusion.

Section 4 describes how TLA⁺, TLAPS, and the Toolbox are used on realistic examples. Their relation to other proof systems is discussed in Section 5. A concluding section summarizes what we have done.

2 Modeling Peterson’s Algorithm In TLA⁺

Peterson’s algorithm is a classic, very simple two-process mutual exclusion algorithm. We specify the algorithm in TLA⁺ and prove that it satisfies mutual exclusion, meaning that no two processes are in their critical sections at the same time.

2.1 From PlusCal To TLA⁺

We will write Peterson’s algorithm in the PlusCal algorithm language. To do so, we have the Toolbox create an empty TLA⁺ module. We name the two processes 0 and 1, and we define an operator *Not* so that *Not*(0) = 1 and *Not*(1) = 0:

$$\text{Not}(i) \triangleq \text{IF } i = 0 \text{ THEN } 1 \text{ ELSE } 0$$

The PlusCal code for Peterson’s algorithm is shown in Figure 1; it appears in a comment in the TLA⁺ module.⁶ The **variables** statement declares the variables

⁶ The figure shows the pretty-printed version of PlusCal code and TLA⁺ formulas. As an example of how they are typed, here is the ASCII version of the **variables** declaration: `variables flag = [i \in {0, 1} |-> FALSE], turn = 0;`

```

--algorithm Peterson {
  variables flag = [i ∈ {0, 1} ↦ FALSE], turn = 0;
  process (proc ∈ {0, 1}) {
    a0: while (TRUE) {
      a1:   flag[self] := TRUE;
      a2:   turn := Not(self);
      a3a:  if (flag[Not(self)]) {goto a3b} else {goto cs} ;
      a3b:  if (turn = Not(self)) {goto a3a} else {goto cs} ;
      cs:   skip; \* critical section
      a4:   flag[self] := FALSE;
    } \* end while
  } \* end process
} \* end algorithm

```

Fig. 1. Peterson’s algorithm in PlusCal.

and their initial values. For example, the initial value of *flag* is an array such that $flag[0] = flag[1] = \text{FALSE}$. (Mathematically, an array is a function; the TLA⁺ notation $[x \in S \mapsto e]$ is similar to a lambda expression.) To specify a multiprocess algorithm, it is necessary to specify what its atomic actions are. In PlusCal, an atomic action consists of the execution from one label to the next. With this brief explanation, the reader should be able to figure out what the code means.

The PlusCal translator, accessible through a Toolbox menu, generates a TLA⁺ specification from the PlusCal code of the algorithm. Figure 2 gives the generated TLA⁺ translation.⁷ The heart of the TLA⁺ specification consists of the initial predicate *Init*, which describes the initial state, and the next-state relation *Next*, which describes how the state can change. The PlusCal compiler adds a variable *pc*, which explicitly records the control state of each process. For example, control in process *i* is at *cs* iff $pc[i]$ equals the string “cs”.

Given the PlusCal code, the meaning of formula *Init* in the figure is straightforward. The formula *Next* is a predicate on old-state/new-state pairs. Unprimed variables refer to the old state and primed variables to the new state. Formula *Next* is the disjunction of the two formulas *proc*(0) and *proc*(1), and each *proc*(*self*) is the disjunction of seven formulas—one for each label in the body of the **process**. The formula *a0*(*self*) specifies the state change performed by process *self* executing an atomic action starting at label *a0*, and similarly for the other six labels. The reader should be able to figure out the meaning of the TLA⁺ notation and of formula *Next* by comparing these seven definitions with the corresponding PlusCal code.

The temporal formula *Spec* is the complete specification. It is satisfied by a behavior iff the behavior starts in a state satisfying *Init* and each of its

⁷ For clarity of presentation, we have simplified the translation slightly by “in-lining” a definition. The proof we develop works for the unmodified translation if we add a global declaration that causes the definition to be expanded throughout the proof.

VARIABLES $flag, turn, pc$
 $vars \triangleq \langle flag, turn, pc \rangle$
 $Init \triangleq \wedge flag = [i \in \{0, 1\} \mapsto \text{FALSE}]$
 $\wedge turn = 0$
 $\wedge pc = [self \in \{0, 1\} \mapsto \text{"a0"}]$
 $a0(self) \triangleq \wedge pc[self] = \text{"a0"}$
 $\wedge pc' = [pc \text{ EXCEPT } ![self] = \text{"a1"}]$
 $\wedge \text{UNCHANGED } \langle flag, turn \rangle$
 $a1(self) \triangleq \wedge pc[self] = \text{"a1"}$
 $\wedge flag' = [flag \text{ EXCEPT } ![self] = \text{TRUE}]$
 $\wedge pc' = [pc \text{ EXCEPT } ![self] = \text{"a2"}]$
 $\wedge turn' = turn$
 $a2(self) \triangleq \wedge pc[self] = \text{"a2"}$
 $\wedge turn' = \text{Not}(self)$
 $\wedge pc' = [pc \text{ EXCEPT } ![self] = \text{"a3a"}]$
 $\wedge flag' = flag$
 $a3a(self) \triangleq \wedge pc[self] = \text{"a3a"}$
 $\wedge \text{IF } flag[\text{Not}(self)]$
 $\quad \text{THEN } pc' = [pc \text{ EXCEPT } ![self] = \text{"a3b"}]$
 $\quad \text{ELSE } pc' = [pc \text{ EXCEPT } ![self] = \text{"cs"}]$
 $\wedge \text{UNCHANGED } \langle flag, turn \rangle$
 $a3b(self) \triangleq \wedge pc[self] = \text{"a3b"}$
 $\wedge \text{IF } turn = \text{Not}(self)$
 $\quad \text{THEN } pc' = [pc \text{ EXCEPT } ![self] = \text{"a3a"}]$
 $\quad \text{ELSE } pc' = [pc \text{ EXCEPT } ![self] = \text{"cs"}]$
 $\wedge \text{UNCHANGED } \langle flag, turn \rangle$
 $cs(self) \triangleq \wedge pc[self] = \text{"cs"}$
 $\wedge pc' = [pc \text{ EXCEPT } ![self] = \text{"a4"}]$
 $\wedge \text{UNCHANGED } \langle flag, turn \rangle$
 $a4(self) \triangleq \wedge pc[self] = \text{"a4"}$
 $\wedge flag' = [flag \text{ EXCEPT } ![self] = \text{FALSE}]$
 $\wedge pc' = [pc \text{ EXCEPT } ![self] = \text{"a0"}]$
 $\wedge turn' = turn$
 $proc(self) \triangleq a0(self) \vee a1(self) \vee a2(self) \vee a3a(self) \vee a3b(self)$
 $\quad \vee cs(self) \vee a4(self)$
 $Next \triangleq \exists self \in \{0, 1\} : proc(self)$
 $Spec \triangleq Init \wedge \square [Next]_{vars}$

Fig. 2. A pretty-printed version of the TLA⁺ translation, slightly simplified.

steps (pairs of successive states) either satisfies $Next$ or else leaves the values of the three variables $flag$, $turn$, and pc unchanged.⁸ The \square is the ordinary *always* operator of linear-time temporal logic, and $[Next]_{vars}$ is an abbreviation for $Next \vee \text{UNCHANGED } vars$, where $\text{UNCHANGED } vars$ is an abbreviation for $vars' = vars$ and priming an expression means priming all the variables that occur in it.

2.2 Validation Through Model Checking

Before trying to prove correctness of the algorithm, we use TLC, the TLA⁺ model checker, to check it for errors. We first instruct the Toolbox to have TLC check for “execution errors”.⁹ What are type errors in typed languages are one source of execution errors in TLA⁺.

The Toolbox runs TLC on a model of a TLA⁺ specification. A model usually assigns particular values to specification constants, such as the number N of processes. It can also restrict the set of states explored, which is useful if the specification allows an infinite number of reachable states. For this trivial example, there are no constants to specify and only 146 reachable states. TLC finds no execution errors.

We next check if the algorithm actually satisfies mutual exclusion. Since we made execution of the critical section an atomic action, mutual exclusion means that the two processes never both have control at label cs . Mutual exclusion therefore holds iff the following predicate *MutualExclusion* is an invariant of the algorithm—meaning that it is true in all reachable states:

$$MutualExclusion \triangleq (pc[0] \neq \text{“cs”}) \vee (pc[1] \neq \text{“cs”})$$

TLC reports that the algorithm indeed satisfies this invariant. Peterson’s algorithm is so simple that TLC has checked that all possible executions satisfy mutual exclusion. For more interesting algorithms that have an infinite set of reachable states, TLC is no longer able to exhaustively verify all executions, and correctness must be proved deductively. Still, TLC is very valuable for catching errors in the algorithm or its formal model: the effort required for running TLC is incomparably lower than that for writing a formal proof.

The following section describes a deductive correctness proof of Peterson’s algorithm in TLA⁺. Proofs of more interesting algorithms follow the same basic structure, but they are longer and require more complicated reasoning. Section 4 describes how TLA⁺ proofs scale to larger algorithms.

⁸ “Stuttering steps” that leave all variables unchanged are allowed in order to make refinement simple [11].

⁹ The translation is a temporal logic formula, so there is no obvious definition of an execution error. An execution error occurs in a behavior if whether or not the behavior satisfies the formula is not specified by the semantics of TLA⁺—for example, because the semantics do not specify whether or not 0 equals FALSE.

THEOREM $Spec \Rightarrow \Box MutualExclusion$
 ⟨1⟩1. $Init \Rightarrow Inv$
 ⟨1⟩2. $Inv \wedge [Next]_{vars} \Rightarrow Inv'$
 ⟨1⟩3. $Inv \Rightarrow MutualExclusion$
 ⟨1⟩4. QED

Fig. 3. The high-level proof.

3 Proving Mutual Exclusion For Peterson’s Algorithm

The assertion that Peterson’s algorithm implements mutual exclusion is formalized in TLA⁺ as:

THEOREM $Spec \Rightarrow \Box MutualExclusion$

The standard method of proving this invariance property is to find an inductive invariant Inv that implies $MutualExclusion$. An inductive invariant is one that is true in the initial state and whose truth is preserved by the next-state relation. TLA⁺ proofs are hierarchically structured and are generally written top-down. The top level of this invariance proof is shown in Figure 3. Step ⟨1⟩2 asserts that the truth of Inv is preserved by the next-state relation.

Each proof in the hierarchy ends with a QED step that asserts the goal of that proof, the QED step for the top level asserting the statement of the theorem. We usually write the QED step’s proof first. This QED step follows easily from ⟨1⟩1, ⟨1⟩2, and ⟨1⟩3 by propositional logic and the following two temporal-logic proof rules:

$$\frac{I \wedge [N]_v \Rightarrow I'}{I \wedge \Box [N]_v \Rightarrow \Box I} \qquad \frac{P \Rightarrow Q}{\Box P \Rightarrow \Box Q}$$

However, TLAPS does not yet handle temporal reasoning, so we omit the proof of the QED step. When temporal reasoning is added to TLAPS, we expect it easily to check such a trivial proof.

To continue the proof, we must define the inductive invariant Inv . Figure 4 defines Inv to be the conjunction of two formulas. The first, $TypeOK$, is a “type-correctness” invariant, asserting that the values of all variables are elements of

$$\begin{aligned} TypeOK &\triangleq \wedge pc \in [\{0, 1\} \rightarrow \{ \text{“a0”}, \text{“a1”}, \text{“a2”}, \text{“a3a”}, \text{“a3b”}, \text{“cs”}, \text{“a4”} \}] \\ &\quad \wedge turn \in \{0, 1\} \\ &\quad \wedge flag \in [\{0, 1\} \rightarrow \text{BOOLEAN}] \\ I &\triangleq \forall i \in \{0, 1\} : \\ &\quad \wedge pc[i] \in \{ \text{“a2”}, \text{“a3a”}, \text{“a3b”}, \text{“cs”}, \text{“a4”} \} \Rightarrow flag[i] \\ &\quad \wedge pc[i] \in \{ \text{“cs”}, \text{“a4”} \} \Rightarrow \wedge pc[Not(i)] \notin \{ \text{“cs”}, \text{“a4”} \} \\ &\quad \quad \wedge pc[Not(i)] \in \{ \text{“a3a”}, \text{“a3b”} \} \Rightarrow turn = i \\ Inv &\triangleq TypeOK \wedge I \end{aligned}$$

Fig. 4. The inductive invariant.

the “right” set. (The expression $[S \rightarrow T]$ is the set of all functions whose domain is S and whose range is a subset of T .) In an untyped logic like that of TLA^+ , almost any inductive invariant must assert type correctness. The second conjunct, I , is the interesting one that explains why Peterson’s algorithm implements mutual exclusion.

There is no point trying to prove that a formula is an inductive invariant if TLC can show that it’s not even an invariant. So, we first run TLC to test if Inv is an invariant. In the simple case of Peterson’s algorithm, TLC can check not only that it is an invariant, but that it is an inductive invariant. We check that Inv is an inductive invariant of $Spec$ by checking that it is an (ordinary) invariant of the specification $Inv \wedge \Box[Next]_{vars}$, obtained from $Spec$ by replacing the initial condition by Inv . In most real examples, TLC can at best check an inductive invariant on a tiny model—one that is too small to gain any confidence that it really is an inductive invariant. However, TLC can still often find simple errors in an inductive invariant.

We now prove steps $\langle 1 \rangle 1$ – $\langle 1 \rangle 3$. We can prove them in any order; let us start with $\langle 1 \rangle 1$. We expect this step to follow easily from the definitions of $Init$ and Inv and simple properties of sets and functions. TLAPS knows about sets and functions, but it does not expand definitions unless directed to do so. (In complex proofs, automatically expanding definitions often leads to formulas that are too big for provers to handle.) We assert that the step follows from simple math and the definitions of $Init$ and Inv by writing the following leaf proof immediately after the step:

BY DEF $Init, Inv$

We then tell the Toolbox to run TLAPS to check this proof. It does so and reports that the prover failed to prove the following obligation:

```

ASSUME NEW VARIABLE flag,
        NEW VARIABLE turn,
        NEW VARIABLE pc
PROVE  ( $\wedge$  flag = [i \in {0, 1}  $\rightarrow$  FALSE]
         $\wedge$  turn = 0
         $\wedge$  pc = [self \in {0, 1}  $\rightarrow$  a0])
         $\Rightarrow$  TypeOK  $\wedge$  I

```

This obligation is exactly what TLAPS’s back-end provers are trying to prove. They are given no other facts. In particular, the provers know nothing about $TypeOK$ and I , so they obviously can’t prove the obligation. We have to tell TLAPS also to use the definitions of $TypeOK$ and I . We do that by making the obvious change to the BY proof, after which TLAPS easily proves the step.

Step $\langle 1 \rangle 3$ is proved the same way, by simply expanding the definitions of $MutualExclusion$, Inv , I , $TypeOK$, and Not . We next try the same technique on $\langle 1 \rangle 2$. A little thought shows that we have to tell TLAPS to expand all the definitions in the module up to and including the definition of $Next$, except for the definition of $Init$. However, when we direct TLAPS to prove the step, it fails to do so, reporting a 65-line proof obligation.

```

⟨1⟩2.  $Inv \wedge [Next]_{vars} \Rightarrow Inv'$ 
  ⟨2⟩1. SUFFICES ASSUME  $Inv, Next$ 
        PROVE  $Inv'$ 
  ⟨2⟩2.  $TypeOK'$ 
  ⟨2⟩3.  $I'$ 
  ⟨2⟩4. QED

```

Fig. 5. The top-level proof of ⟨1⟩2.

TLAPS uses Zenon and Isabelle as its default backend provers, first trying Zenon and then trying Isabelle if Zenon fails to find a proof. However, TLAPS also includes an SMT solver backend [18] that is capable of handling larger “shallow” proof obligations—in particular, ones that do not contain significant quantifier reasoning. We instruct TLAPS to use the SMT back end when proving the current step by writing

BY SMT DEF ...

The SMT back end translates the proof obligation to SMT-LIB [3], the standard input language for different SMT solvers, and calls an SMT solver (CVC3 by default) to try to prove the resulting formula. CVC3 proves step ⟨1⟩2 in a few seconds. Variants of the SMT back end translate to the native input languages of Yices and Z3, which sometimes perform better than does CVC3 using the standard SMT-LIB translation.

For sufficiently complicated examples, an SMT solver will not be able to prove inductive invariance as a single obligation. The proof will have to be hierarchically decomposed. To illustrate how this is done, we now write a proof of ⟨1⟩2 that can be checked using only the Zenon and Isabelle back-end provers.

Step ⟨1⟩2 and its top-level proof appear in Figure 5. The first step in the proof of an implication like this would normally be:

```

⟨2⟩1. SUFFICES ASSUME  $Inv, [Next]_{vars}$ 
        PROVE  $Inv'$ 

```

This step asserts that to prove the current goal, which is step ⟨1⟩2, it suffices to assume that Inv and $[Next]_{vars}$ are true and prove Inv' . The step also changes the goal of the rest of the level-2 proof to Inv' and allows the assumptions Inv and $[Next]_{vars}$ to be used in the rest of the proof. This step’s assertion is obviously true, and TLAPS will check the one-word leaf proof OBVIOUS. However, the proof of Figure 5 does something a little different.

Since the assumption $[Next]_{vars}$ equals $Next \vee \text{UNCHANGED } vars$, it leaves two cases to be proved: (i) $Next$ is true and (ii) all variables are unchanged, so their primed values equal their unprimed values. The proof in the second case is trivial, and TLAPS should have no trouble checking it. In Figure 5, the assumption in the SUFFICES statement is $Next$ rather than $[Next]_{vars}$, so the remainder of the proof only has to consider case (i). To show that it suffices to prove Inv' under this stronger assumption, the proof of that SUFFICES step has to prove case (ii).

The remainder of the level-2 proof is straightforward. Since Inv equals $TypeOK \wedge I$, the goal Inv' is the conjunction of the two formulas $TypeOK'$ and I' . We therefore decompose the proof by proving each conjunct separately. The proof of the QED step is simply

BY $\langle 2 \rangle 2, \langle 2 \rangle 3$ DEF Inv

Observe that we have to tell TLAPS exactly what facts to use as well as what definitions to expand.

We next prove $\langle 2 \rangle 1$ – $\langle 2 \rangle 3$. Zenon proves $\langle 2 \rangle 1$ when the definitions of $vars$, Inv , $TypeOK$, and I are expanded. Note that the definition of $Next$ is not needed. To prove $\langle 2 \rangle 2$ and $\langle 2 \rangle 3$, we need to use the definition of $Next$ —that is, with all definitions expanded down to TLA⁺ primitives—as well as the definition of Inv . We also have to use the assumption that Inv and $Next$ are true, introduced by step $\langle 2 \rangle 1$. This leads us to try the following proof for $\langle 2 \rangle 2$.

BY $\langle 2 \rangle 1$ DEF $Inv, TypeOK, Next, proc, a0, a1, a2, a3a, a3b, cs, a4, Not$

Zenon fails on this proof, but Isabelle succeeds. However, both Zenon and Isabelle fail on the corresponding proof of $\langle 2 \rangle 3$ (which requires also using the definition of I). To prove it (with only Zenon and Isabelle), we need one more level of proof. That level appears in Figure 6, which contains the complete proof of the theorem.

Since priming a formula means priming all variables in it, the goal I' has the form $\forall i \in \{0, 1\} : exp(i)'$. A standard way to prove this formula is by \forall -introduction: we introduce a new variable, say j , we assume $j \in \{0, 1\}$, and we prove $exp(j)'$. TLA⁺ provides a notation for naming subexpressions of a definition. With that notation, the expression $exp(j)$ is written $I!(j)$. This leads us to begin the proof of $\langle 2 \rangle 3$ with the SUFFICES step $\langle 3 \rangle 1$ of Figure 6 and its simple proof.

The assumption $Next$ (introduced by $\langle 2 \rangle 1$) equals $\exists self \in \{0, 1\} : proc(self)$. A standard way to use such an assumption is by \exists -elimination: we pick some value of $self$ such that $proc(self)$ is true. That is what step $\langle 3 \rangle 2$ does, naming the value i .

We simplified our task to proving $I!(j)'$ instead of I' , using $proc(i)$ instead of $Next$, which eliminates two quantifiers. However, Zenon and Isabelle still cannot prove the goal in a single step. The usual way to decompose the proof that process i preserves an invariant is to show that each separate atomic action of process i preserves the invariant. In mathematical terms, $proc(i)$ is the disjunction of the seven formulas $a0(i), \dots, a4(i)$, each describing one of the process's atomic action. We can decompose the proof by considering each of the seven formulas as a separate case.

While this is the usual procedure, Peterson's algorithm is simple enough that it is not necessary. Instead, we just have to help the back-end provers by splitting the proof into the two cases of $i = j$ and $i \neq j$. The reader can see how this is done in Figure 6. Observe that in the proof of CASE statement $\langle 3 \rangle 3$, the name $\langle 3 \rangle 3$ refers to the case assumption $i = j$. There is no explicit use of $\langle 3 \rangle 1$

THEOREM $Spec \Rightarrow \square MutualExclusion$

$\langle 1 \rangle 1. Init \Rightarrow Inv$
 BY DEF $Init, Inv, I, TypeOK$

$\langle 1 \rangle 2. Inv \wedge [Next]_{vars} \Rightarrow Inv'$
 $\langle 2 \rangle 1.$ SUFFICES ASSUME $Inv, Next$
 PROVE Inv'
 BY DEF $vars, Inv, TypeOK, I$

$\langle 2 \rangle 2. TypeOK'$
 BY $\langle 2 \rangle 1$ DEF $Inv, TypeOK, Next, proc, a0, a1, a2, a3a, a3b, cs, a4, Not$

$\langle 2 \rangle 3. I'$
 $\langle 3 \rangle 1.$ SUFFICES ASSUME NEW $j \in \{0, 1\}$
 PROVE $I!(j)'$
 BY DEF I

$\langle 3 \rangle 2.$ PICK $i \in \{0, 1\} : proc(i)$
 BY $\langle 2 \rangle 1$ DEF $Next$

$\langle 3 \rangle 3.$ CASE $i = j$
 BY $\langle 2 \rangle 1, \langle 3 \rangle 2, \langle 3 \rangle 3$ DEF $Inv, I, TypeOK, proc, a0, a1, a2, a3a, a3b, cs, a4, Not$

$\langle 3 \rangle 4.$ CASE $i \neq j$
 BY $\langle 2 \rangle 1, \langle 3 \rangle 2, \langle 3 \rangle 4$ DEF $Inv, I, TypeOK, proc, a0, a1, a2, a3a, a3b, cs, a4, Not$

$\langle 3 \rangle 5.$ QED
 BY $\langle 3 \rangle 3, \langle 3 \rangle 4$

$\langle 2 \rangle 4.$ QED
 BY $\langle 2 \rangle 2, \langle 2 \rangle 3$ DEF Inv

$\langle 1 \rangle 3. Inv \Rightarrow MutualExclusion$
 BY DEF $MutualExclusion, Inv, I, TypeOK, Not$

$\langle 1 \rangle 4.$ QED
 PROOF OMITTED

Fig. 6. The complete proof.

because a NEW assumption in an ASSUME is used by default in all proofs in the assumption's scope. The same is true of the formula $i \in \{0, 1\}$ asserted by the PICK step. (This is a pragmatic choice in the design of TLAPS, based on the observation that such facts are used so often.)

4 Writing Real Proofs

We have described how one writes and checks a TLA⁺ proof of a tiny example. Several larger case studies have been carried out using the system. These include verifications of Byzantine Paxos [16], the Memoir security architecture [19], and the lookup and join protocols of the Pastry algorithm for maintaining a distributed hashtable over a peer-to-peer network [17]. TLA⁺ and TLAPS, with its Toolbox interface, provide a number of features to help manage the complexity of large proofs.

4.1 Hierarchical Proofs And The Proof Manager

The most important aid in writing large proofs is TLA⁺'s hierarchical and declarative proof language, where intermediate proof obligations are stated explicitly. While declarative proofs are more verbose than standard tactic scripts, they are easier to understand and maintain because the information on what is currently being proved is available at each point. Hierarchical proofs enable a user to keep decomposing a complex proof into smaller steps until the steps become provable by one of the back-end provers.

In logical terms, proof steps correspond to natural-deduction sequents whose validity must be established in the current context, which consists of constant and variable symbols, assumptions, and already-established facts. The Proof Manager tracks the context, which is modified by non-leaf proof steps. For leaf proof steps, it computes a corresponding proof obligation, sends it to the backend provers, and records the status of the step's proof (succeeded, failed, canceled by the user, or omitted).

Because proof obligations are independent of one another, users can develop proofs in any order and work on the proof of a step independently of the state of the proof of any other step. This permits them to concentrate on the part of a planned proof that is most likely to be wrong and require changes to other parts of the proof. The Toolbox makes it easy to instruct TLAPS to check the proof of everything in a file, of any single theorem, or of any single step. It displays every obligation whose proof fails or is taking too long; in the latter case the user can cancel the proof. Clicking on the obligation shows the part of the proof that generated it.

A linear presentation, as in Figure 6, is unsuitable for reading or writing large proofs. The Toolbox's editor helps reading and writing large TLA⁺ proofs, providing commands that show or hide particular subproofs. Commands to hide a proof or view just its top level aid in reading a proof. A command that is particularly useful when writing a subproof is one that hides all preceding steps that cannot be used in that subproof because of their positions in the hierarchy.

TLA⁺'s hierarchical proofs provide a much more powerful mechanism for structuring complex proofs than the conventional approach using lemmas. In a TLA⁺ proof, each step with a non-leaf proof is effectively a lemma. One typical 1100-line invariance proof [16] contains 100 such steps. A conventional linear proof with 100 lemmas would be impossible to read.

4.2 Fingerprinting: Tracking The Status Of Proof Obligations

During proof development, a user repeatedly modifies the proof structure or changes details of the specification. Rerunning the backend provers on a sizable proof takes time. By default, TLAPS does not reprove an obligation that it has already proved—even if the proof has been reorganized and the step that generated it has been moved, or if the step was removed from the proof and reinserted in a later version. It can also show the user the impact of a change by indicating which parts of the existing proof must be reproved.

The Proof Manager computes a *fingerprint* of every obligation, which it stores, along with the obligation’s status, in a separate file. Technically, a proof obligation is canonically represented as a lambda term, with bound variables replaced by de Bruijn indices [7] so their actual names in the TLA⁺ proof are irrelevant. The context is minimized by erasing symbols and hypotheses that are not used in the step. The fingerprint is a hash-consed version of the resulting representation, which is therefore insensitive to structural modifications of the proof context that do not affect the obligation’s logical validity.

The Toolbox displays the proof status of each step, indicating by color whether the step has been proved or some obligation in its proof has failed or has been omitted. Looking up an obligation’s status takes little time, so the user can tell TLAPS to re-prove a step or a theorem even if only a small part of the proof has changed; TLAPS will recognize any obligation that has not changed and will not attempt to prove it anew. There is also a check-status command that displays the proof status without actually launching any proofs.

An incident that occurred in the Byzantine Paxos proof reveals the advantages of our method of writing proofs. The third author wrote the safety proof primarily as a way of debugging TLAPS, spending a total of several weeks over several months on it. Later, when writing a paper about the algorithm, he discovered that it did not satisfy the desired liveness property, so it had to be modified. He changed the algorithm, fixed minor bugs found by TLC, and reproved the safety property—all in a day and a half, with about 12 hours of actual work. He was able to do it that fast because of the hierarchical proof structure, TLAPS’s fingerprinting mechanism (about 3/4 of the proof obligations in the new proof had already been proved), and the Toolbox’s aid in managing the proof. Even with this help, it would have taken much longer if the bugs found by TLC had had to be found and fixed while writing the proof.

5 Related Work

The TLA⁺ proof system is a platform for interactive proofs; it is mainly intended for verifying concurrent and distributed algorithms. In contrast to most interactive proof assistants [24], TLAPS is based on a declarative proof language that is independent of any specific proof backend. TLA⁺ proofs indicate what facts are needed to prove a certain result, but they do not specify precisely how the backend provers should use these facts (e.g., for forward or backward chaining, or for rewriting). This can frustrate users who are intimately familiar with the inner workings of a particular prover and know how to apply them in the most efficient manner. However, declarative proofs are less dependent on specific backend provers and less sensitive to changes in what is to be proved.

We write complex proofs by hierarchically structuring their logic. The graphical user interface provides commands that support hierarchical proofs by allowing a user to zoom in on the current context and by supporting non-linear proof development. Although some other interactive proof systems such as Mizar [21] and Isabelle/Isar [22] also offer hierarchical proofs, to the best of our knowledge

these systems do not provide the Toolbox’s abilities to use that structure to aid in reading and writing proofs and to prove individual steps in any order—facilities that we find crucial in developing and managing large proofs. The only other proof assistants that we know to offer a mechanism comparable to our fingerprinting facility is the KIV system [2].

The Rodin toolset supporting the Event-B formal method [1] is related to TLAPS: Event-B and TLA^+ are both based on Zermelo-Fraenkel set theory, both emphasize refinement as a way to structure formal developments, and Rodin and TLAPS support proofs of safety properties, using different backend provers. Unlike with Event-B models, the structure of TLA^+ specifications is not fixed: any TLA^+ formula can be considered as a system specification or a property, and TLAPS does not impose a structure on invariant or refinement proofs. The interactive prover of the Rodin toolset provides commands to invoke tactics rather than being based on a declarative proof language.

Tools and methods for program verification such as VCC [6] or Why [10] are based on the generation of precise verification condition according to program instructions, for which they provide powerful automatic provers. The main user interaction consists in choosing suitable programs annotations. These tools therefore target the verification of executable code rather than high-level descriptions of algorithms.

6 Conclusion

We have used Peterson’s algorithm as an example to present the main constructs of the TLA^+ proof language and the interaction of a user with the TLA^+ proof system. That algorithm was chosen because it is well known and simple—so simple that all the high-level steps of the mutual exclusion proof (Figure 3) could have been proved by TLAPS with leaf proofs. We explained in Section 4 why TLA^+ proofs scale to more complex algorithms and systems that we do not expect any prover to handle automatically. The hierarchical structure of the proof language is essential for giving users flexibility in designing their proof structure, and it ensures that individual proof steps are independent of one another. The fingerprinting mechanism of TLAPS makes use of this independence by storing previously proved results and retrieving them, even when they appear in a different context.

While not illustrating the entire proof language [15], Peterson’s algorithm does show its main features. Every step states an assertion. Leaf proofs immediately prove a step, citing the necessary definitions, facts, and assumption. Non-leaf proofs consist of another level of proof steps that end with a QED step establishing the assertion of the preceding level. This basic structure is oriented towards forward-style proofs, but the judicious use of backward chaining (indicated by the keyword SUFFICES) can make proofs more readable. Some features of the proof language that do not appear in the proof of Figure 6 are constructs for providing a witness to prove an existentially quantified formula, introducing

local definitions, and specifying facts that can be used by the backend provers even when they are not explicitly mentioned.

Different proof techniques, such as resolution, tableau methods, rewriting engines, and SMT solvers offer complementary strengths. Future versions of TLAPS will probably add new back-end provers and may launch several provers in parallel for proving each obligation. Adding a new back end mainly involves writing a translation from TLA^+ to the input language of the prover. Such translations can be complex, and there is a legitimate concern about their soundness as well as about the soundness of the back ends themselves. For back ends that can produce proof traces, TLAPS provides the option to certify the traces within Isabelle. Proof trace certification has been implemented for Zenon, and we plan to implement it for other back ends including SMT solvers. Still, it is much more likely that a proof is meaningless because of an error in the formula we are proving than because of an error in a back end. Moreover, soundness also depends on parts of the proof manager. Users who do not trust its fingerprinting mechanism can disable it and reprove the entire proof or any part of it. However, the proof manager carries out some transformations, such as replacing $(a + b)'$ by $(a' + b')$, that are critical for soundness.

We cannot overstate how important it is that TLAPS is integrated with the other TLA^+ tools—especially the TLC model checker. Checking putative invariants and assertions with TLC on finite instances of a specification is much more productive than discovering errors during the proof. Users check the exact same specifications that appear in their proofs. Less obvious is how useful it is that TLC can evaluate TLA^+ formulas. When verifying a system, we don't want to prove well-known mathematical facts; we want to assume them. However, it is easy to make a mistake in formalizing even simple mathematics, and assuming the truth of an incorrect formula can lead to an incorrect proof. TLC can usually check a formula on a large enough model to make us confident that our formalization of a correct mathematical result is indeed correct.

We are actively developing TLAPS. Our main short-term objective is to add support for temporal reasoning. It is not obvious how best to extend natural deduction to temporal logic. We have designed an approach involving two forms of sequents, expressed with two forms of the ASSUME/PROVE statement having different semantics, that we think will work well. We will know if it does only after implementing it and using it. We also plan to improve support for standard TLA^+ data structures such as sequences.

References

1. J.-R. Abrial, M. Butler, S. Hallerstede, T. S. Hoang, F. Mehta, and L. Voisin. Rodin: an open toolset for modelling and reasoning in Event-B. *STTT*, 12(6): 447–466, 2010.
2. M. Balsler, W. Reif, G. Schellhorn, K. Stenzel, and A. Thums. Formal system development with KIV. In T. Maibaum, editor, *3rd Intl. Conf. Fundamental Approaches to Software Engineering*, volume 1783, pages 363–366. Springer, Berlin, Germany, 2000.

3. C. Barrett, L. de Moura, S. Ranise, A. Stump, and C. Tinelli. The smt-lib initiative and the rise of smt. In *Hardware and Software: Verification and Testing*, Lecture Notes in Computer Science.
4. R. Bonichon, D. Delahaye, and D. Doligez. Zenon : An extensible automated theorem prover producing checkable proofs. In N. Dershowitz and A. Voronkov, editors, *14th Intl. Conf. Logic for Programming, Artificial Intelligence, and Reasoning (LPAR 2007)*, volume 4790 of *LNCS*, pages 151–165. Springer, Yerevan, Armenia, 2007.
5. K. Chaudhuri, D. Doligez, L. Lamport, and S. Merz. Verifying safety properties with the tla⁺ proof system. In J. Giesl and R. Hähnle, editors, *5th Intl. Joint Conf. Automated Reasoning (IJCAR 2010)*, volume 6173 of *Lecture Notes in Computer Science*, pages 142–148. Springer, Edinburgh, UK, 2010.
6. E. Cohen, M. Dahlweid, M. Hillebrand, D. Leinenbach, M. Moskal, T. Santen, W. Schulte, and S. Tobies. VCC: A practical system for verifying Concurrent C. In *22nd Intl. Conf. Theorem Proving in Higher Order Logics (TPHOLs 2009)*, volume 5674 of *LNCS*, pages 23–42. Springer, Munich, Germany, 2009.
7. N. G. de Bruijn. Lambda calculus notation with nameless dummies. a tool for automatic formula manipulation with application to the church-rosser theorem. *Indagationes Mathematicae*, 34:381–392, 1972.
8. L. M. de Moura and N. Bjørner. Z3: An efficient smt solver. In *TACAS*, pages 337–340, 2008.
9. B. Dutertre and L. D. Moura. The yices smt solver. Technical report, SRI, 2006.
10. P. Herms, C. Marché, and B. Monate. A certified multi-prover verification condition generator. In R. Joshi, P. Müller, and A. Podelski, editors, *4th Intl. Conf. Verified Software: Theories, Tools, Experiments (VSTTE 2012)*, volume 7152 of *LNCS*, pages 2–17. Springer, Philadelphia, PA, 2012.
11. L. Lamport. What good is temporal logic? In R. E. A. Mason, editor, *Information Processing 83: Proceedings of the IFIP 9th World Congress*, pages 657–668. IFIP, North-Holland, Paris, Sept. 1983.
12. L. Lamport. How to write a proof. *American Mathematical Monthly*, 102(7): 600–608, Aug. 1993.
13. L. Lamport. *Specifying Systems: The TLA⁺ Language and Tools for Hardware and Software Engineers*. Addison-Wesley, 2003.
14. L. Lamport. The PlusCal algorithm language. In M. Leucker and C. Morgan, editors, *6th Intl. Coll. Theoretical Aspects of Computing (ICTAC 2009)*, volume 5684 of *Lecture Notes in Computer Science*, pages 36–60. Springer, Kuala Lumpur, Malaysia, 2009.
15. L. Lamport. TLA⁺2: A preliminary guide. Draft manuscript, Nov. 2011. <http://research.microsoft.com/users/lamport/tla/tla2-guide.pdf>.
16. L. Lamport. Byzantizing paxos by refinement. Available at <http://research.microsoft.com/en-us/um/people/lamport/pubs/web-byzpxos.pdf>, 2011.
17. T. Lu, S. Merz, and C. Weidenbach. Towards verification of the Pastry protocol using TLA⁺. In R. Bruni and J. Dingel, editors, *Formal Techniques for Distributed Systems (FORTE 2011)*, volume 6722 of *LNCS*, pages 244–258. Springer, Reykjavik, Iceland, 2011.
18. S. Merz and H. Vanzetto. Automatic verification of TLA⁺ proof obligations with SMT solvers. In N. Bjørner and A. Voronkov, editors, *18th Intl. Conf. Logic for Programming, Artificial Intelligence and Reasoning*, LNCS. Springer, Mérida, Venezuela, 2012. To appear.

19. B. Parno, J. R. Lorch, J. R. Douceur, J. Mickens, and J. M. McCune. Memoir: Practical state continuity for protected modules. In *Proc. IEEE Symp. on Security and Privacy*, pages 379–394. IEEE, 2011.
20. G. L. Peterson. Myths about the mutual exclusion problem. *Inf. Process. Lett.*, 12(3):115–116, 1981.
21. A. Trybulec. Mizar. In Wiedijk [24], pages 20–23.
22. M. Wenzel and L. C. Paulson. Isabelle/Isar. In Wiedijk [24], pages 41–49.
23. M. Wenzel, L. C. Paulson, and T. Nipkow. The Isabelle framework. In O. A. Mohamed, C. Muñoz, and S. Tahar, editors, *21st Intl. Conf. Theorem Proving in Higher Order Logics (TPHOLs 2008)*, volume 5170 of *LNCS*, pages 33–38. Springer, Montreal, Canada, 2008.
24. F. Wiedijk, editor. *The Seventeen Provers of the World*, volume 3600 of *LNCS*. Springer, 2006.